
Analyse, simulation d’une porte XOR CMOS à 6 transistors 
  

 

J’invite le lecteur à consulter le site pour des informations complémentaires.  
 

Page d’accueil du site Internet :    d’autres pdf, sur différents sujets : 
page d'accueil                    liste des PDF  

 

 

La porte XOR, (OU exclusif) est un opérateur de base en logique combinatoire. Il en existe 

plusieurs schémas à transistors. Le but de ce travail est de vérifier -et de caractériser- par 

des simulations Pspice, le fonctionnement d’une porte XOR CMOS à 6 transistors. Dans 

cet article, on connecte également un montage analogique en sa sortie. 

 

 

1) Porte XOR à base de portes logiques. 

 

Rappelons les schémas des opérateurs logiques en technologie CMOS : 

 

 

 

   
NOT 

 

 

NAND 

 

AND 

 

OR 

 

2 transistors 4 transistors 6 transistors 4 transistors 

 
 

L'opérateur XOR (X = A  B) a pour équation A\ B + A B\ . 

On peut donc le fabriquer classiquement avec des portes de base :  

NOT, AND, OR. 

 

Le bilan des opérateurs nécessaires à la réalisation de l'opérateur 

XOR totalise 20 transistors. 

 
Porte XOR à base d’inverseurs, 

de ET, de OU. 

 
 

 

 

La même équation logique peut être réalisée par des opérateurs 

NAND. 

 

Le bilan fait alors état de 16 transistors. 

 
Porte XOR à base d’inverseurs, 

et de NAND. 

 

Il est également possible d’assembler des transistors de façon judicieuse tout en respectant la fonction 

logique du OU exclusif.  

 

Parmi les différentes solutions, nous allons examiner finement la porte XOR à 6 transistors. 

  

http://serge.dusausay.free.fr/
http://serge.dusausay.free.fr/ss/plan2.html

2) Porte XOR à 6 transistors. 

 

Il est proposé ci-contre un schéma qui n'exploite pas les schémas 

classiques des opérateurs logiques.  

Dans cette technologie, l’alimentation VDD  est égale à 3,3 V. 

 

On peut remarquer : 
 

Porte XOR 
 

 

- il y a 2 inverseurs logiques (MP1, MN1 d'une part, MP3, MN3 d'autre part), 
 

- l'association MP2, MN2 ne forme pas un inverseur logique (mais une porte de 

transmission), 
 

- sur le plan logique, la fonction XOR est commutative. Mais, sur le plan analogique, 

ce schéma n'est pas symétrique. 

 
Porte XOR 

 

- Le signal interne XN (pour XNOR, XOR complémenté) n'assure pas des niveaux logiques "de qualité". 

L'étage de sortie permet de réduire ce défaut : l'inverseur entre XN et X permet de remettre en forme des 

signaux dégradés. 

 

 

Pour vérifier le fonctionnement de la porte XOR, on place, sur les 

entrées A et B,  des signaux de test comme indiqué ci-contre : 

 

On a :   niveau logique "0" = 0 V niveau logique "1" = 3,3 V. 
 

Signaux de test 

 

La fréquence des signaux est fixée arbitrairement à 2 MHz, les temps de montée et descente à 2 ns. 

Dans la netlist (fichier XOR.cir), le décalage est de 125 ns, soit T/4, ce qui revient à /2.  

 

La porte "XOR" est définie dans un sous-circuit. 
 
XOR à 6 Transistors 
* fichier XOR.cir 
* Xor entrées A, B sortie X 

* transistors AMSD : 
.model MODN nmos (Level=1 Kp=110u Vto=0.68 Cgdo=0.5n Cgso=3.2n lambda=0.01) 
.model MODP pmos (Level=1 Kp=40u  Vto=-0.7 Cgdo=0.5n Cgso=3.2n lambda=0.01) 

* circuit : 
Vdd vdd 0 dc 3.3V ; alim 
 
* test XOR 
VinA A 0 pulse ( 0 3.3 0n 2n 2n 0.25u 0.5u ) ; signal 2 MHz  
VinB B 0 pulse ( 0 3.3 0.125u 2n 2n 0.25u 0.5u ) ; décalé 
Xxor A B X OUEX 
 
.subckt OUEX A B X  
Valim Vdd 0 DC=3.3 
Mp2 XN A BN Vdd MODP W=1.6u L=0.35u ; DGSB 
Mn2 XN A B 0 MODN W=1u L=0.35u ; DGSB 
Xinv1 B BN INVCMOS 
Xinv2 XN X INVCMOS 
.ends 
 
.subckt INVMF in out  
Valim Vdd 0 DC=3.3 
Mp out in vdd vdd MODP W=6u L=0.35u ; W/L grand 
Mn out in 0 0 MODN W=0.35u L=4u ; W/L petit 
.ends 

 
.subckt INVCMOS in out  
Valim Vdd 0 DC=3.3 
Mp out in vdd vdd MODP W=1.6u L=0.35u ; DGSB 
Mn out in 0 0 MODN W=1u L=0.35u ; DGSB 
.ends 
 
* filtre de boucle  entree UD (noeud X) sortie UF 
*R1 X UF 130k 
*R2 UF UFM 14.45k  
*CB UFM 0 40p   
* 
* filtre de sortie 
*RF UF USD 289k 
*CF USD 0 2p 
 
.TRAN 0.1n 1u 0 0.1n ; reponse temporelle 1 
*.TRAN 1n 100u 0 1n ; reponse temporelle 2 
.PROBE 
.END 

Netlist, prête à simuler 

Après simulation, on représente ci-dessous les chronogrammes décalés, pour une lecture plus facile : 

De haut en bas : A, B, X.  

 
 

- Ce tableau synthétique représente les chronogrammes des signaux A, B, X. On indique l’état logique 

correspondant aux potentiels (« 1 » pour 3,3 V et « 0 » pour 0 V). 
 

V(A) 1 1 0 0 1 1 0 

V(B) 0 1 1 0 0 1 1 

V(X) 1 0 1 0 1 0 1 

 

- on reconnait la table de vérité de la fonction XOR : l'un ou l'autre, mais pas les 2. 

 

- il est judicieux d'observer également les signaux internes BN et XN :   

BN est l’inversion de B, sans défaut, hormis un très léger pic lors d'une commutation (ici à 500 ns) 

mais XN est manifestement dégradé. 

 

 
En effet, les niveaux de XN sont, selon la combinaison des entrées A, B : 
 

A B XN Remarque 

1 0 0 V  

0 0 3,3 V  

0 1 703 mV Est interprété comme 0, par l’inverseur qui suit. 

1 1 2,6 V Est interprété comme 1, par l’inverseur qui suit. 

 

Sur le plan fonctionnel, en ne se limitant qu’aux signaux d’entrées A, B, et de la sortie X, ce montage à 6 

transistors réalise bien la fonction OU Exclusif. 

 

Une des applications du XOR est le comparateur de phase de certaines boucles à verrouillage de phase. 

C’est ce qui est traité au paragraphe suivant. Généralement, la sortie du comparateur de phase est 

connectée à un filtre de type passe bas. Nous allons simuler l’association XOR et filtre passe bas passif.  

           Time

0s 100ns 200ns 300ns 400ns 500ns 600ns 700ns 800ns

8+V(A) 4+V(B) V(X)

0V

5V

10V

           Time

0s 100ns 200ns 300ns 400ns 500ns 600ns 700ns 800ns

V(Xxor.XN) 4+ V(Xxor.BN)

0V

4.0V

8.0V

3) Cas pratique d’utilisation du XOR : le comparateur de phase 

 

Nous allons tout d’abord étudier le filtre passe bas connecté sur la porte XOR, appelé, dans le cas d’une 

PLL, filtre de boucle. 

 

a) Le filtre de boucle seul :  

Le filtre est 
pτ1

pτ1
K  

(p)U

(p)U
F(p)

2

1
f

D

F




 , avec τ1 = R2 C et τ2 = ( R1 + R2 ) C. 

On donne : R1 = 130 kΩ, R2 = 14,45 kΩ, C = 40 pF. 

On déduit :  
 

Filtre 

τ1 = R2 C =  14,45 103 x  40 10-12 = 0,578 µs   d'où 1/τ1 = 1,73 Mrad/s    

τ2 = ( R1 + R2 ) C = (130 + 14,45) 103 x 40 10-12  = 5,778  µs  d'où 1/τ2 = 173 krad/s. 

Kf = 1. 

 

Dans la PLL qui exploitera ce comparateur et ce filtre, on connecte un 

autre filtre, appelé filtre de sortie. 

 

On donne : RF = 289 kΩ, CF = 2 pF. 

Sa fréquence de coupure est 1/(2 π RF CF) = 275 kHz. 

 

 
 

Les 2 filtres en cascade 

L'impédance de sortie du filtre de boucle (R1 // (R2+1/jC) est faible devant l'impédance d'entrée du filtre 

de boucle (RF + 1/jCF). On peut donc faire l’hypothèse que la fonction de transfert F(p) est inchangée 

par la présence du filtre de sortie.  

 

Si on veut s'en convaincre, il suffit de faire 2 réponses harmoniques du filtre de boucle : une fois seul, une 

fois avec le filtre de sortie.  
les 2 filtres 
* fichier filtre.cir 

* circuit : 
VAC X 0 AC=1 
* filtre de boucle  chargé 
R1 X UF 130k 
R2 UF UFM 14.45k  
CB UFM 0 40p   
* filtre de sortie 
RF UF USD 289k 
CF USD 0 2p 

* filtre de boucle à vide 
R12 X UF2 130k 

R22 UF2 UFM2 14.45k  
CB2 UFM2 0 40p   
.AC DEC 100 1k 10Meg 
.PROBE 
.END 

 
 

Interprétation : 
 

La réponse du filtre seul (rouge), ou du filtre chargé (bleu) sont quasiment confondues : la mise en 

cascade du filtre de sortie ne modifie pas la réponse du filtre de boucle. 

L'atténuation maximale arrive pour f >1/2τ1  = 275 kHz, et vaut alors 0,1. (-20 dB). 
  

           Frequency

1.0KHz 3.0KHz 10KHz 30KHz 100KHz 300KHz 1.0MHz 3.0MHz 10MHz

DB((V(UF2))) DB((V(UF)))

-20

-10

0

b) Association XOR et filtres :  
 

Pour vérifier le fonctionnement en comparateur de phase, on recommence l'essai, avec cette fois-ci, le 

XOR chargé par le filtre de boucle, lui-même connecté au filtre de sortie (il suffit, dans le fichier 

XOR.cir, de dé-commenter les 5 lignes concernées). On valide la ligne ".TRAN 1n 100u 0 1n". 

 

- Le signal issu du XOR n’est pas dégradé car l'impédance d'entrée des éléments est élevée, grâce à la 

résistance R1 = 130 kΩ qui est en série avec les autres éléments. On mesure, à quelques mV près, des 

niveaux standards 3,3 V ; 0 V.  

Remarque : sans conditions initiales sur la tension aux bornes des condensateurs, il faut laisser passer 

quelques dizaines de microsecondes pour avoir des courbes stabilisées. 

 

- On visualise UF et USD : 

 
En UF, le signal est formée d'une composante continue de ≈ 1,65 V, (représentative de la valeur moyenne, 

car le signal en sortie du XOR est de rapport cyclique 50 %) + une variation.  

Un zoom sur cette variation montre le niveau bas environ 1,48 V, le niveau haut environ 1,8 V, soit une 

variation de 0,32 V crête à crête environ. 

 

En USD, cette variation est atténuée. Le filtre passe-bas de sortie, qui coupe à 275 kHz, atténue d’un 

coefficient 4/0,275 = 14,5 ,  le fondamental à 4 MHz.  

On visualise la tension en sortie de ce dernier filtre : elle est une tension constante de ≈ 1,65 V, avec une 

variation triangulaire quasi négligeable de 33 mV crête à crête.  

 

Conclusion 

 

Utilisé en comparateur de phase, le XOR doit délivrer une tension image du déphasage entre les signaux 

appliqués en entrée. C’est ce qui est réalisé par l’association du XOR à un filtre passe bas qui prend la 

valeur moyenne. 

 

On peut ainsi idéaliser la réponse de ce détecteur de phase par la caractéristique 

ci-contre : autour de la position centrale, on a <UD> = Kd ER, avec la 

convention usuelle sinus, cosinus, pour la définition du retard ER. 
 

Remarques :  
 

- le signe de Kd dépend de l'affectation des signaux sur les entrées A et B. 
 

- cette réponse idéalisée suppose des signaux parfaits en A, B, X. 

 
<UD> = f (ER) 

 

Les niveaux haut et bas étant quasiment 0 V et 3,3 V, on a : Kd = 3,3 V /  = 1,05 V /rad. 

Si les signaux sont en quadrature, <UD> = 1,65 V.  

Dans la convention adoptée, l'écart de phase = 0. D'où la fonction de transfert tracée. 

           Time

99.0us 99.1us 99.2us 99.3us 99.4us 99.5us 99.6us 99.7us 99.8us 99.9us100.0us

V(UF) V(USD)

1.5V

1.6V

1.7V

1.8V

